


Application Note		資料作成 081127	資料番号 an2k-002
テーマ	RS-485 接続温湿度計の記録と取り出し その1。 USBメモリとCUnetを使った2通りのデータログ方法です。		
使用機器	MPC-2100, MPC-Cunet, MRS-MCOM, USB-CUnet タッチパネル:パナソニック 電工 GT11, 温湿度計:CHNO HN-CHNS, USBメモリ(市販) Windows PC(MS-EXCEL)		

■機器構成



■概要


- ・ 2 台の温湿度計のデータを一定間隔で USB メモリに CSV 形式で保存します。同時に USB-CUnet 経由で PC (A) の EXCEL のデータシートに直接入力してオンタイムでグラフを更新します。
- ・ USB メモリのデータは計測終了後に PC (B) に差し替えて EXCEL に入力します。
- ・ タッチパネルに現在の温度、湿度、日時を表示します。



温湿度計(床)




タッチパネル画面



手前から MPC-2100, MRS-MCOM, MPC-CUnet

## ■RS-485 結線



MRS-MCOM には Fail Safe 回路が組み込まれているので外付け回路はいりません。

## ■温湿度計の通信フォーマット

- ・温湿度計に対する測定データ要求フォーマット

ENQ	機器番号	STX	RPV01	ETX	CR	LF
-----	------	-----	-------	-----	----	----

- ・温湿度計からの応答フォーマット

ACK	機器番号	STX	APV01	=	年	月	日	時	分	温度	湿度	ETX	CR	LF
-----	------	-----	-------	---	---	---	---	---	---	----	----	-----	----	----

- ・温湿度計の応答例 (応答文字列は固定長です。温湿度計のカレンダーは無通電で 2001/01/01 になります)

(文字数) 0 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19 20 21 22 23 24 25  
 (HEX) 06 30 31 02 41 50 56 30 31 3D 32 30 30 31 2C 20 31 2C 30 31 2C 32 30 2C 34 32  
 (ASCII) AK 0 1 SX A P V 0 1 = 2 0 0 1 , SP 1 , 0 1 , 2 0 , 4 2

(文字数) 26 27 28 29 30 31 32 33 34 35 36 37 38 39 40 41 42 43 44 45 46 47 48 49 50  
 (HEX) 2C 30 2C 30 2C 30 2C 20 20 32 32 2E 33 2C 30 2C 20 20 36 31 2E 37 03 0D 0A  
 (ASCII) , 0 , 0 , 0 , SP SP 2 2 . 3 , 0 , SP SP 6 1 . 7 EX CR LF

## ■MPC プログラム

```

CUNET 4 4 8 /* CUNet 初期化
MEWNET 38400 2 /* タッチパネル通信
S_MBK &H20 8 /* タッチパネルページ切り替え
CNFG# 5 RS485 "9600b7pes1NONE" /* MRS-MCOM ch5 RS-485 モード
OFF 70000 /* START SW オフ
FOR I=100 TO 131 /* タッチパネル使用エリア初期化
  S_MBK 0 I
NEXT I
S_MBK 0 200 /* Write Count
S_MBK 0 300 /* タッチパネル Msg "Push START"

*START
WAIT SW(70000)=1 /* タッチパネル START SW (オルタネート)
S_MBK 1 300 /* タッチパネル Msg "Logging"
FILENAME$="test.csv" /* ファイル名
USB_DEL FILENAME$ /* USB メモリ 既存ファイル削除
APPEND FILENAME$ /* USB メモリ 追記モードでオープン

FORMAT "0000/00/00" /* 文字列書式
DT$=HEX$(DATE(0))
S_MBK DT$ 124 10 /* タッチパネル 年月日表示
ST$="時間, 温度 1, 湿度 1, 温度 2, 湿度 2, "+DT$+"¥n" /* scv の表題
PRINT# USB ST$ /* USB メモリ 書き込み
DELAY=100 /* read 間隔 必要に応じて加減
WRITE_COUNT=0
  
```

```

DO
  CH$="01" /* 温湿度計 ch1
  GOSUB *GET_DATA
  RES1$=RES$
  S_MBK RES1$ 100 9 /* タッチパネル 温度,湿度表示
  OUT t 2032~Int /* CUnet に温度書き込み
  OUT h 2034~Int /* CUnet に湿度書き込み
  TIME DELAY /* この温湿度計には必要みたい

  CH$="02" /* 温湿度計 ch2
  GOSUB *GET_DATA
  RES2$=RES$
  S_MBK RES2$ 108 9
  OUT t 2036~Int
  OUT h 2038~Int

  FORMAT "00:00:00"
  TM$=HEX$(TIME(0))
  S_MBK TM$ 116 8 /* タッチパネル 時間表示
  ST$=TM$+", "+RES1$+", "+RES2$+"¥n"
  PRINT ST$
  IF (TIME(0)&&HFF)==0 THEN /* 秒が00 なら書き込み=60 秒間隔
    INC WRITE_COUNT /* カウンタインクリメント
    S_MBK WRITE_COUNT 200~Wrd /* タッチパネル カウンタ値表示
    PRINT "WRITE TO USB-MEM" WRITE_COUNT
    PRINT# USB ST$ /* USBメモリ 書き込み
    WAIT (TIME(0)&&HOF) !=0 /* 秒!=0 待ち
  END_IF
  TIME DELAY /* この温湿度計には必要みたい

  IF SW(70000)==0 THEN /* タッチパネル STOP SW 押下なら
    CLOSE USB
    S_MBK 2 300 /* タッチパネル Msg "USB Close"
    GOTO *START
  END_IF
  HH=TIME(0)/&H10000 /* MPC の日時を CUnet に書き込む
  MM=TIME(0)/&H100
  MM=MM&&HFF
  SS=TIME(0)&&HFF
  OUT HH 2040
  OUT MM 2041
  OUT SS 2042

LOOP

*GET_DATA
PRINT# 5 CHR$(5) CH$ CHR$(2) "RPV01" CHR$(3) "¥r¥n" /* 温湿度計にコマンド送信
INPUT# 5 a$ /* 温湿度計データ受信

/*表示して温度湿度以外は読み飛ばす例
/* PRINT VAL(a$) VAL(0) VAL(0) VAL(0) VAL(0) VAL(0) VAL(0) VAL(0) VAL(0)
/* PRINT t=VAL(10) VAL(0) h=VAL(10)

/*表示せず読み飛ばし
dummy=VAL(a$)
FOR I=1 to 9
  dummy=VAL(0)
NEXT I
t=VAL(10) /* 温度値取得
dummy=VAL(0) /* 読み飛ばし
h=VAL(10) /* 湿度値取得


FORMAT "00.0" /* 書式=整数 2 桁、小数 1 桁
RES$=STR$(t) /* 温度文字列
FORMAT "00.0"
RES$=RES$+", "+STR$(h) /* 温度文字列,湿度文字列

RETURN

```

## ■PC(A)のEXCEL VBA プログラム

リアルタイムにデータを取得して EXCEL のデータシートに追加&グラフ化します。



### ◆標準モジュール

usbuncunet.dll コマンド宣言、グラフ表示、データ取得

```

Declare Function usb_open Lib "usbuncunet.dll" () As Long
Declare Function fw_vid Lib "usbuncunet.dll" () As Long
Declare Function fw_pid Lib "usbuncunet.dll" () As Long
Declare Function fw_ver Lib "usbuncunet.dll" () As Long
Declare Function dll_ver Lib "usbuncunet.dll" () As Long
Declare Sub init_cunet Lib "usbuncunet.dll" (ByVal sa As Long, ByVal ow As Long, ByVal en As Long)
Declare Function cunet_in Lib "usbuncunet.dll" (ByVal adr As Long, ByVal siz As Long) As Long
Declare Sub cunet_out Lib "usbuncunet.dll" (ByVal dat As Long, ByVal adr As Long, ByVal siz As Long)

```

```
Public Declare Sub Sleep Lib "kernel32" (ByVal dwMilliseconds As Long)
```

```

Public Const CuChr = 1 '1byte 符号無
Public Const CuInt = 2 '2Byte 符号付
Public Const CuWrd = 4 '2Byte 符号無
Public Const CuLng = 8 '4Byte 符号付

```

```
Global RunFlg
```

```
Public Sub graph_create()
```

```

nChart = ActiveSheet.ChartObjects.Count
If nChart > 0 Then '既存グラフがあるときは消す
    For cnt = 1 To nChart
        ActiveSheet.ChartObjects(1).Activate '消すと count は減算されるので常に 1
        ActiveChart.ChartArea.Select
        ActiveWindow.Visible = False
        ActiveChart.Parent.Delete
    
```

```

        ' Selection.Delete
    Next
End If

Columns("A:E").Select
Charts.Add
ActiveChart.ChartType = xlLine
ActiveChart.SetSourceData Source:=Worksheets("Sheet1").Range("A1:E2"), PlotBy _
:=xlColumns
ActiveChart.Location Where:=xlLocationAsObject, Name:="Sheet1"
With ActiveChart
    .HasTitle = True
    .ChartTitle.Characters.Text = Date
    .Axes(xlCategory, xlPrimary).HasTitle = False
    .Axes(xlValue, xlPrimary).HasTitle = False
End With

End Sub

Public Sub get_data(r, c)
    Cells(r, c) = Time$
    Cells(r, c + 1) = cunet_in(2032, CuInt) / 10 'ch1 温度
    Cells(r, c + 2) = cunet_in(2034, CuInt) / 10 'ch1 湿度
    Cells(r, c + 3) = cunet_in(2036, CuInt) / 10 'ch2 温度
    Cells(r, c + 4) = cunet_in(2038, CuInt) / 10 'ch2 湿度

    ActiveSheet.TextBox1.Text = "Latest Data (row=" + CStr(r) + ") " _
        + Time$ + Chr(13) + Chr(10) _
        + " 温度 1:" + Format(Cells(r, c + 1), "00.0°C") _
        + " 湿度 1:" + Format(Cells(r, c + 2), "00.0%") + Chr(13) + Chr(10) _
        + " 温度 2:" + Format(Cells(r, c + 3), "00.0°C") _
        + " 湿度 2:" + Format(Cells(r, c + 4), "00.0%") _

End Sub

◆Sheet1
Start ボタンでデータシート・グラフ・CUnet 初期化、60 秒間隔でデータを取得して cell とグラフに追加
Stop ボタンで停止

Private Sub CommandButton1_Click()

    Cells.Clear ' 既存データクリア
    Columns("A:A").Select
    Selection.NumberFormatLocal = "hh:mm:ss" ' セルの書式=時刻
    Columns("B:E").Select
    Selection.NumberFormatLocal = "0.0_" ' セルの書式=数値
    Range("A1").Select

    If usb_open <> 1 Then
        Cells(1, 7) = "USB OPEN ERROR"
        Exit Sub
    End If
    CommandButton1.Enabled = False
    CommandButton2.Enabled = True
    CommandButton2.Activate
    DoEvents

    init_cunet 0, 4, 8 ' CUnet 初期化

    Cells(1, 1) = "時刻"
    Cells(1, 2) = "温度 1"
    Cells(1, 3) = "湿度 1"
    Cells(1, 4) = "温度 2"
    Cells(1, 5) = "湿度 2"

    Label1.Caption = "FW " + CStr(fw_ver) + " DLL " + CStr(dll_ver)

    ' グラフ初期化のダミー書き込み
    get_data 2, 1

    graph_create ' グラフ作成

    RunFlg = 1
    cr = 2 ' row

```

```

Do
  If (Int(Timer) Mod 60) = 0 Then
    nChart = ActiveSheet.ChartObjects.Count
    If nChart > 0 Then
      For cnt = 1 To nChart
        ActiveSheet.ChartObjects(1).Activate
        ActiveChart.ChartArea.Select
        ActiveWindow.Visible = False

        get_data cr, 1

        ActiveWindow.Visible = True
        ActiveChart.SetSourceData Source:=Worksheets("Sheet1").Range("A1:E" + CStr(cr)), PlotBy _
          :=xlColumns

      Next

    End If
    cr = cr + 1
    st = Timer
    st = st + 2
    Do
      If Timer > st Then Exit Do
      Sleep 1
      DoEvents
    Loop
  End If

  DoEvents
  If RunFlg = 0 Then Exit Do
  Sleep 1
  ' MPC の時計
  HH = Hex(cunet_in(2040, CUBYTE))
  MM = Hex(cunet_in(2041, CUBYTE))
  SS = Hex(cunet_in(2042, CUBYTE))
  Label2.Caption = "MPC TIME " + Format(HH, "00:") + Format(MM, "00:") + Format(SS, "00")

Loop
CommandButton2.Enabled = False

End Sub

Private Sub CommandButton2_Click()
  RunFlg = 0
  CommandButton1.Enabled = True
  CommandButton1.Activate
  Range("A1").Select
End Sub

```

## ■USBメモリの記録

測定作業終了後、USBメモリーをPC(B)に移してデータを処理します。

1分間隔 535個のデータで約16kByteでした。

## ◆TEST.CSVの内容

時間, 温度1, 湿度1, 温度2, 湿度2, 2008/11/19

09:07:00, 22.9, 23.7, 17.3, 36.9

09:08:00, 23.1, 23.5, 17.3, 37.2

09:09:00, 23.3, 24.3, 17.4, 37.4

09:10:00, 23.3, 24.9, 17.5, 37.5

09:11:00, 23.4, 24.7, 17.5, 37.4

09:12:00, 23.4, 24.9, 17.7, 37.4

09:13:00, 23.5, 24.5, 17.8, 37.0

(中略)

17:56:00, 22.9, 20.4, 20.9, 24.8

17:57:00, 22.9, 20.6, 20.9, 26.9

17:58:00, 23.0, 20.5, 20.9, 24.8

17:59:00, 22.9, 20.2, 20.9, 24.6

18:00:00, 23.0, 20.1, 20.9, 26.5

18:01:00, 23.1, 19.9, 21.0, 24.5

◆PC(B)のEXCELでTEST.CSVを読み込み、グラフウィザードで描画しました。

エアコン暖房の作動状況、“頭熱足寒”状態などがわかります。

